

Sample queries in data.bnf.fr

1.1 Discover the data model

What properties were used to describe the resource <http://data.bnf.fr/ark:/12148/cb11933798p>

```
SELECT DISTINCT ?p ?o WHERE {
  <http://data.bnf.fr/ark:/12148/cb11933798p> ?p ?o.
}
```

1.2 Information on a person or an organization

1.2.1 Information on an author

Example: All the authors' biographical dates (birthday, year of birth and year of death) and names

```
SELECT ?auteur ?jour ?date1 ?date2 ?nom where {
  ?auteur foaf:birthday ?jour.
  ?auteur bio:birth ?date1.
  ?auteur bio:death ?date2.
  OPTIONAL {?auteur foaf:name ?nom}
}
ORDER BY (?jour)
```

Example 2: All the information about Victor Hugo and all the labels associated with this person

```
SELECT DISTINCT ?nom_complet ?nom ?prenom ?forme_retenue
?formes_rejetees ?pays ?langue ?sexe ?anniversaire ?date_naissance
?lieu_naissance ?date_mort ?lieu_mort ?periode_activite ?domaine_activite
?bio WHERE {
  <http://data.bnf.fr/ark:/12148/cb11907966z#foaf:Person> foaf:gender ?sexe;
  <http://rdvocab.info/ElementsGr2/countryAssociatedWithThePerson>
  ?pays;
  <http://rdvocab.info/ElementsGr2/languageOfThePerson> ?langue;
  <http://rdvocab.info/ElementsGr2/dateOfBirth> ?date_naissance;
  <http://rdvocab.info/ElementsGr2/placeOfBirth> ?lieu_naissance;
  foaf:birthday ?anniversaire;
  <http://rdvocab.info/ElementsGr2/dateOfDeath> ?date_mort;
  <http://rdvocab.info/ElementsGr2/placeOfDeath> ?lieu_mort;
  <http://rdvocab.info/ElementsGr2/fieldOfActivityOfThePerson>
  ?domaine_activite;
  <http://rdvocab.info/ElementsGr2/biographicalInformation> ?bio;
  foaf:name ?nom_complet;
  foaf:familyName ?nom;
  foaf:givenName ?prenom.
  <http://data.bnf.fr/ark:/12148/cb11907966z> skos:altLabel ?formes_rejetees;
  skos:prefLabel ?forme_retenue
  OPTIONAL {<http://data.bnf.fr/ark:/12148/cb11907966z#foaf:Person>
  <http://rdvocab.info/ElementsGr2/periodOfActivityOfThePerson>
  ?periode_activite}}
```



Example 3: All the composer's birthdays with their name and birth and death years.

```
SELECT ?auteur ?jour ?date1 ?date2 ?nom where {
  ?auteur foaf:birthday ?jour.
  ?doc bnfroles:r220 ?auteur.
  ?auteur <http://rdvocab.info/ElementsGr2/dateOfBirth> ?date1.
  ?auteur <http://rdvocab.info/ElementsGr2/dateOfDeath> ?date2.
  OPTIONAL {?auteur foaf:name ?nom}
}
ORDER BY (?jour)
```

Example 4: For each author, find the pages containing the documents about that author

```
SELECT DISTINCT ?auteur ?documents_a_propos_de WHERE {
  ?doc dcterms:creator ?uri_auteur .
  ?documents_a_propos_de dcterms:subject ?uri_auteur .
  ?uri_auteur foaf:name ?auteur
  FILTER REGEX (?documents_a_propos_de, ".*studies.*")
}
```

**This query only retrieves pages from data.bnf.fr (with 'studies' in their URI)*

```
SELECT DISTINCT ?auteur ?documents_a_propos_de WHERE {
  ?doc dcterms:creator ?uri_auteur .
  ?documents_a_propos_de dcterms:subject ?uri_auteur .
  ?uri_auteur foaf:name ?auteur
  FILTER (!REGEX (?documents_a_propos_de, "http://data.bnf.fr/*.*))
}
```

**Conversely, this query only retrieves pages that are outside data.bnf.fr*

1.2.2 Relationships between authors and works

Example: find all the works by Victor Hugo with their title

```
SELECT DISTINCT ?work ?title WHERE
{
  <http://data.bnf.fr/ark:/12148/cb11907966z> foaf:focus ?person.
  ?work dcterms:creator ?person ;
    rdfs:label ?title .
}
```

1.3 Information on a work

Example 1: Find all the title forms of the Roman de la Rose and the associated note

```
SELECT DISTINCT ?titre_forme_internationale_francais ?formes_rejetees ?note_associee
WHERE {
  <http://data.bnf.fr/ark:/12148/cb166125510> skos:altLabel ?formes_rejetees;
  skos:prefLabel ?titre_forme_internationale_francais;
  skos:editorialNote ?note_associee
}
```



Example 2: find the list of the editions of a given work (here, les Fleurs du mal)

```
SELECT DISTINCT ?title ?date ?editeur WHERE {
  <http://data.bnf.fr/ark:/12148/cb11947965f> foaf:focus ?Work .
  ?Work rdarelationshps:expressionOfWork ?Expression .
  ?Manif rdarelationshps:expressionManifested ?Expression.
  OPTIONAL{?Manif dcterms:date ?date}
  OPTIONAL{?Manif dcterms:title ?title}
  OPTIONAL{?Manif dcterms:publisher ?editeur}
}
```

1.4 Relationships between authors and expressions: roles played by an author on documents

Example 1: Find translators of Italian works in French

```
SELECT DISTINCT ?notice ?name WHERE {
  ?Work dcterms:language <http://id.loc.gov/vocabulary/iso639-2/ita> .
  ?Exp bnfroles:r680 ?traducteur ;
    dcterms:language <http://id.loc.gov/vocabulary/iso639-2/fre>.
  ?Work rdarelationshps:expressionOfWork ?Exp .
  ?traducteur foaf:name ?name.
  ?notice foaf:focus ?traducteur.
}
```

Example 2: Find the first and last name of the photographers

```
SELECT DISTINCT ?Photographe ?Prenom ?Nom WHERE
{
  ?expression bnfroles:r530 ?Photographe .
  ?Photographe a foaf:Person.
  OPTIONAL {?Photographe foaf:givenName ?Prenom.}
  OPTIONAL {?Photographe foaf:familyName ?Nom.}
}
```

1.5 Subjects (RAMEAU)

Example 1: For a given RAMEAU subject heading, retrieve all direct narrower terms and, if any, retrieve the latter terms' own narrower terms

```
SELECT DISTINCT ?original_rameau ?prefLabel ?uri_a ?label_a ?uri_b ?label_b
WHERE {
  ?original_rameau skos:prefLabel ?prefLabel ;
    skos:narrower ?uri_a .
  MINUS {?original_rameau foaf:focus ?focus .}
  ?uri_a skos:prefLabel ?label_a .
```



```
OPTIONAL {
?uri_a skos:narrower ?uri_b .
?uri_b skos:prefLabel ?label_b .
}
}
```

Example 2: Documents about the subject « Fencing » (URI <http://data.bnf.fr/ark:/12148/cb11931273g>)

```
SELECT ?doc {
?doc dcterms:subject <http://data.bnf.fr/ark:/12148/cb11931273g>.
}
```

Example 3: Publications about photography

```
SELECT DISTINCT ?Edition ?title WHERE {
?Edition a frbr-rda:Manifestation;
dcterms:subject <http://data.bnf.fr/ark:/12148/cb1193113t>;
dcterms:title ?title.
}
```

1.6 Identifiers

Exemple 1: For a given ISNI, get the first and last name of the corresponding person (here Alexandre Dumas)

```
SELECT DISTINCT ?nom ?prenom WHERE
{
?person isni:identifierValid "0000000121012885" ;
foaf:focus ?identity.
?identity foaf:familyName ?nom;
foaf:givenName ?prenom.
}
```

Example 2: Find the ARK identifier of a record from its “FRBNF” (traditional record number) (here, the title « les Travailleurs de la mer »)

```
SELECT DISTINCT ?idArk WHERE
{
?idArk bnf-onto:FRBNF "11992081"^^xsd:integer.
}
```

Example 3: Find the work(s) and author’s name from a given ISBN

Property used: bnf-onto:isbn

Example: find « Extension du domaine de la lutte » from the ISBN "2-7028-4777-3".

```
SELECT DISTINCT ?work ?title ?name WHERE
{
?work rdfs:label ?title;
dcterms:creator ?creator.
}
```



```
?work rdarelationshps:expressionOfWork ?expression.
?manifestation rdarelationshps:expressionManifested ?expression.
?manifestation bnf-onto:isbn "2-7028-4777-3".
?creator foaf:name ?name.}
```

1.7 Vocabulary schemes and reference datasets: musical genres, languages, countries, document types, works for children

Example 1: select all documents with the type « image ».

```
SELECT ?image where {
?image dc:type <http://purl.org/dc/dcmitype/StillImage>.
} Limit 100
```

Example 2: Count the number of works per musical genre; start the list with the most represented genres

```
SELECT DISTINCT ?genre ?label COUNT (?work) AS ?nbWork
WHERE {
?work mo:genre ?genre.
?genre skos:prefLabel ?label }
ORDER BY DESC (?nbWork)
```

Example 3: Retrieve all works that have been adapted for children and the corresponding adapted work

```
SELECT ?uri ?oeuvre WHERE {
?manifestation bnf-onto:ouvrageJeunesse "true"^^xsd:boolean ;
rdarelationshps:expressionManifested ?expression ;
rdfs:seeAlso ?uri.
?oeuvre rdarelationshps:expressionOfWork ?expression.
}
```

1.8 Queries on dates

Birthday in the form mm-jj: foaf:birthday

The exact dates of birth and death in the form aaaa-mm-jj: bio:Birth and bio:Death

Only the years of birth and death: bnf-onto:firstYear and bnf-onto:last year.

Example: retrieve all the authors that died before 1924

```
select distinct ?nom ?auteur ?mort where {
?oeuvre dcterms:creator ?auteur .
?auteur rdf:type foaf:Person ;
bio:Death ?mort ;
foaf:name ?nom
FILTER (xsd:integer (?mort) < "1924"^^xsd:integer )
}
ORDER BY DESC (?mort)
```



*does not retrieve years that do not match the pattern of an integer (e.g. uncertain dates do not appear)

Retrieve all authors born before 1500, ordered by year of birth

```
SELECT ?auteur ?naissance where {?auteur bnf-onto:firstYear ?naissance.
?auteur a foaf:Person.
Filter (?naissance < "1500"^^xsd:integer)
}
ORDER BY ASC (?naissance)
```

1.9 Places and geographical records

Find all digitized documents with geographical records as a subject

```
SELECT DISTINCT ?lieu ?doc ?docnum where {
?lieu rdf:type geo:SpatialThing.
?doc dcterms:subject ?concept.
?concept foaf:focus ?lieu.
?doc rdarelationshp:electronicReproduction ?docnum.
}
```

Find all digitized documents with geographical records or geographical themes as a subject

```
SELECT DISTINCT ?lieu ?doc ?docnum where {
{
?lieu rdf:type <http://www.w3.org/2003/01/geo/wgs84_pos#SpatialThing>.
?doc dcterms:subject ?concept.
?concept foaf:focus ?lieu.
?doc rdarelationshp:electronicReproduction ?docnum.}
UNION {
?doc dcterms:subject ?rameau.
?lieu skos:exactMatch ?rameau.}
}
```

Find all geographical records aligned with a corresponding subject heading:

```
SELECT ?lieu ?concept where {
?c a <http://www.w3.org/2003/01/geo/wgs84_pos#SpatialThing>. ?lieu foaf:focus ?c. ?lieu
skos:closeMatch ?concept.
}
```

1.10 Performances

1.11 Links to digitized documents from Gallica

Example 1: find all the digitized documents about a particular subject

Example: digitized photographs (linked with a « photographer » role) from Gallica, with their URL and their title (first 1000 results)

```
SELECT DISTINCT ?Doc ?title ?URLGallica WHERE
{
?Expression bnfroles:r530 ?Creator.
?Doc rdarelationshp:expressionManifested ?Expression.
?Photo rdarelationshp:electronicReproduction ?URLGallica; dcterms:title
```



```
?title.  
}  
LIMIT 1000
```

1.12 Illustration of a page

Example 1: Find all the illustrations available for the page of Charles Baudelaire

```
SELECT DISTINCT ?image WHERE  
{  
<http://data.bnf.fr/ark:/12148/cb118905823#foaf:Person> foaf:depiction  
?image.  
}
```

Example 2: Images from Gallica with the type « portraits » et their corresponding ISNI identifiers, for French authors

```
select ?auteur ?url where {  
?auteur rdf:type skos:Concept ;  
foaf:focus ?person.  
?doc rdarelations:electronicReproduction ?url.  
?doc dcterms:subject ?auteur ;  
dcterms:subject <http://data.bnf.fr/ark:/12148/cb11932843f>.  
?person <http://rdvocab.info/ElementsGr2/countryAssociatedWithThePerson>  
<http://id.loc.gov/vocabulary/countries/fr#>.  
}
```